MQL4 Book 2

Creation of Custom Indicators

When creating a trading strategy a developer often faces the necessity to draw graphically in a security window a certain dependence calculated by a user (programmer). For this purpose MQL4 offers the possibility of creating custom indicators.

Custom Indicator is an application program coded in MQL4; it is basically intended for graphical displaying of preliminarily calculated dependences.

Custom Indicator Structure

Necessity of Buffers

The main principle underlying custom indicators is passing values of indicator arrays to a client terminal (for drawing indicator lines) via exchange buffers.

Buffer is a memory area containing numeric values of an indicator array.

MQL4 standard implies the possibility of drawing up to eight indicator lines using one custom indicator. One indicator array and one buffer are brought into correspondence with each indicator line. Each buffer has its own index. The index of the first buffer is 0, of the second one - 1, and so on, the last one has the index 7. Fig. 115 shows how the information from a custom indicator is passed via buffers to a client terminal for drawing indicator lines.


The general order of building indicator lines is the following:

1. Calculations are conducted in a custom indicator; as a result numeric values are assigned to indicator array elements.

2. Values of indicator array elements are sent to a client terminal via buffers.

3. On the bases of value arrays received from buffers a client terminal displays indicator lines.

Components of a Custom Indicator

Let's analyze a simple custom indicator that shows two lines - one line is build based on maximal bar prices, the second one uses minimal prices.

	Example of a simple custom indicator userindicator.mq4


//--------------------------------------------------------------------
// userindicator.mq4 
// The code should be used for educational purpose only.
//--------------------------------------------------------------------
#property indicator_chart_window    // Indicator is drawn in the main window
#property indicator_buffers 2       // Number of buffers
#property indicator_color1 Blue     // Color of the 1st line
#property indicator_color2 Red      // Color of the 2nd line
 
double Buf_0[],Buf_1[];             // Declaring arrays (for indicator buffers)
//--------------------------------------------------------------------
int init()                          // Special function init()
  {
   SetIndexBuffer(0,Buf_0);         // Assigning an array to a buffer
   SetIndexStyle (0,DRAW_LINE,STYLE_SOLID,2);// Line style
   SetIndexBuffer(1,Buf_1);         // Assigning an array to a buffer
   SetIndexStyle (1,DRAW_LINE,STYLE_DOT,1);// Line style
   return;                          // Exit the special funct. init()
  }
//--------------------------------------------------------------------
int start()                         // Special function start()
  {
   int i,                           // Bar index
       Counted_bars;                // Number of counted bars
//--------------------------------------------------------------------
   Counted_bars=IndicatorCounted(); // Number of counted bars
   i=Bars-Counted_bars-1;           // Index of the first uncounted
   while(i>=0)                      // Loop for uncounted bars
     {
      Buf_0[i]=High[i];             // Value of 0 buffer on i bar
      Buf_1[i]=Low[i];              // Value of 1st buffer on i bar
      i--;                          // Calculating index of the next bar
     }
//--------------------------------------------------------------------
   return;                          // Exit the special funct. start()
  }
//--------------------------------------------------------------------
Let's analyze in details the indicator parts. In any application program written in MQL4 you can indicate setup parameters that provide the correct program servicing by a client terminal. In this example the head program part (see Program Structure) contains several lines with directives #property .

The first directive indicates in what window the client terminal should draw the indicator lines:

#property indicator_chart_window             // Indicator is drawn in the main window
In MQL4 there are two variants of drawing indicator lines: in the main security window and in a separate window. Main window is the window containing a security chart. In this example parameter indicator_chart_window in #property directory indicates that a client terminal should draw indicator lines in the main window.

The next line shows the number of buffers used in the indicator:

#property indicator_buffers 2                // Number of buffers
In the analyzed example two indicator lines are drawn. One buffer is assigned to each buffer, so the total number of buffers is two.

The next lines describe colors of the indicator lines.

#property indicator_color1 Blue              // Color of the 1st line
#property indicator_color2 Red               // Color of the 2nd line
Parameters indicator_color1 and indicator_color2 define color setting for corresponding buffers - in this case for buffers with indexes 0 (Blue) and 1 (Red). Note that figures in parameter names indicator_color1 and indicator_color2 are not buffer indexes. These figures are parts of constant names that are set in accordance with buffers. For each constant color can be set at the discretion of a user.

In the next line indicator arrays are declared:

double Buf_0[],Buf_1[];                      // Declaring arrays (for indicator buffers)
The indicator is intended for drawing two indicator lines, so we need to declare two global one-dimension arrays, one for each line. Names of indicator arrays are up to user. In this case array names Buf_0[] and Buf_1[] are used, in other cases other names can be used, for example, Line_1[],Alfa[], Integral[] etc. It is necessary to declare arrays on a global level, because array elements values must be preserved between calls of the special function start().

The described custom indicator is built on the basis of two special functions -init() and start(). The function init() contains the part of code used on the program only once (see Special functions).

A very important action is performed in the line:

   SetIndexBuffer(0,Buf_0);                  // Assigning an array to a buffer
Using the function SetIndexBuffer() a necessary buffer (in this case with the index 0) is put into correspondence with an array (in this case Buf_0). It means for constructing the first indicator line a client terminal will accept data contained in the array Buf_0 using the zero buffer for it.

Further the line style is defined:

   SetIndexStyle (0,DRAW_LINE,STYLE_SOLID,2);// Line style
For the zero buffer (0) a client terminal should use the following drawing styles: simple line (DRAW_LINE), solid line (STYLE_SOLID), line width 2.

The next two lines contain settings for the second line:

   SetIndexBuffer(1,Buf_1);                 // Assigning an array to a buffer
   SetIndexStyle (1,DRAW_LINE,STYLE_DOT,1); // Line style
Thus, according to the code of the special function init() both indicator lines will be drawn in the main security window. The first one will be a solid blue line with the width 2, the second one is a red dotted line ( STYLE_DOT) of a usual width. Indicator lines can be drawn by other styles as well (see Styles of Indicator Lines).

Calculating Values of Indicator Arrays Elements (Be Attentive)

Values of indicator arrays elements are calculated in the special function start(). To understand correctly the contents of start() code pay attention to the order of indexing bars. The section Arrays describes in details the method of indexing arrays-timeseries. According to this method bar indexing starts from zero. The zero bar is a current yet unformed bar. The nearest bar's index is 1. The next one's is 2 and so on.

As new bars appear in a security window, indexes of already formed (history) bars are changed. The new (current, just formed, rightmost) bar gets the zero index, the one to the left of him (that has just fully formed) gets the index 1 and values of indexes of all history bars are also increased by one.

	The described method of indexing bars is the only one possible for the whole on-line trading system MetaTrader, and it is taken into account when drawing lines using both technical and custom indicators.


It was said earlier that indicator lines are constructed on the basis of numeric information contained in indicator arrays. An indicator array contains information about dots coordinates upon which an indicator line is drawn. And the Y coordinate of each dot is the value of an indicator array element, and X coordinate is the value of an indicator array element index. In the analyzed example the first indicator line is drawn using maximal values of bars. Fig, 116 shows this indicator line (of blue color) in a security window, it is built on the basis of the indicator array Buf_0.

Fig. 116. Correspondence of coordinates of an indicator line to values of an indicator array.

Index value of an indicator array is out by a client terminal into correspondence with a bar index - these index values are equal. It must be also taken into account that the process of constructing indicator lines goes on in real time mode under conditions when in a security window new bars appear from time to time. And all history bars are shifted to the left. To have the indicator line drawn correctly (each line dot above its bar) it must also be shifted together with bars. So there is need (technical need) to re-index an indicator array.

The fundamental difference of an indicator array from a usual array is the following:

	At the moment when a new bar is created, index values of indicator array elements are automatically changed by the client terminal, namely - value of each indicator array index is increased by one and the indicator array size is increased by one element (with a zero index).


For example, the zero bar in Fig. 116 (timeframe H1) has the opening time 6:00. At 7:00 a new bar will appear in the security window. The bar opened at 6:00 will automatically get the index 1. To have the indicator line drawn correctly on this bar, the client terminal will change the index of the indicator array element corresponding to the bar opened at 6:00. In the table in Fig. 116 this element is written in the first line. Together with that indexes of all array elements will be increased by the client terminal by one. An the index of the array element corresponding to the bar opened at 6:00 will get the value 1 (before that it was equal to 0). The indicator array will become larger by one element. The index of a new added element will be equal to 0, the value of this element will be a new value reflecting coordinate of the indicator line on a zero bar. This value is calculated in the special function start() on each tick.

Calculations in the special function start() should be conducted so that no extra actions were performed. Before the indicator is attached to a chart, it does not reflect any indicator lines (because values of indicator arrays are not defined yet). That's why at the first start of the special function start() indicator array values must be calculated for all bars, on which the indicator line should be drawn. In the analyzed example these are all bars present on a chart (the initial calculations can be conducted not for all available bars, but for some last part of history; it is described in further examples). Ar all further starts of the special function start() there is no need to calculate values of indicator array for all bars again. These values are already calculated and are contained in the indicator array. It is necessary to calculate the current value of the indicator line only on each new tick of the zero bar.

For the implementation of the described technology there is a very useful standard function in MQL4 - IndicatorCounted() .

Function IndicatorCounted()

int IndicatorCounted()
This function returns the number of bars that have not changed since the last indicator call.

If the indicator has never been attached to a chart, at the first start() execution the value of Counted_bars will be equal to zero:

   Counted_bars=IndicatorCounted();  // Number of counted bars
It means the indicator array does not contain any element with earlier predefined value, that is why the whole indicator array must be calculated from beginning to end. The indicator array is calculated from the oldest bar to the zero one. Index of the oldest bar, starting from which calculations must be started, is calculated the following way:

   i=Bars-Counted_bars-1;           // Index of the first uncounted
Suppose at the moment of attaching the indicator there are 300 bars in a chart window. This is the value of the predefined variable Bars. As defined earlier, Counted_bars is equal to 0. So, as a result we obtain that i index of the first uncounted bar (the latest one, starting from which calculations should be conducted) is equal to 299.

All values of indicator array elements are calculated in the loop while():

   while(i>=0)                      // Loop for uncounted bars
     {
      Buf_0[i] = High[i];           // Value of 0 buffer on i bar
      Buf_1[i] = Low[i];            // Value of 1st buffer on i bar
      i--;                          // Calculating index of the next bar
     }
While i is within the range from the first uncounted bar (299) to the current one (0) inclusively, values of indicator array elements are calculated for both indicator lines. Note, missing values of indicator array elements are calculated during one (the first) start of the special function start(). During calculations the client terminal remembers elements, for which values were calculated. The last iteration in while() is performed when i is equal to 0, i.e. values of indicator arrays are calculated for the zero bar. When the loop is over, the special function start() finishes its execution and control is passed to the client terminal. The client terminal in its turn will draw all (in this case two) indicator lines in accordance with the calculated values of array elements.

On the next tick start() will be started by the client terminal again. Further actions will depend on the situation (we will continue analyzing the example for 300 bars).

Variant 1. A new tick comes during the formation of the current zero bar (the most common situation).

Fig. 117. The processed tick belongs to the current bar.

Fig. 117 shows two ticks received by the terminal at moments of time t 1 and t 2. The analyzed situation will be the same for both ticks. Let's trace the execution of start() that was launched at the moment t 2. During the execution of the function start() the following line will be executed:

   Counted_bars=IndicatorCounted(); // number of counted bars
IndicatorCounted() will return the value 299, i.e. since the last start() call 299 previous bars were not changed. As a result i index value will be equal to 0 (300-299-1):

   i=Bars-Counted_bars-1;           // Index of the first uncounted
It means in the next while() loop the values of array elements with the zero index will be calculated. In other words, the new position of an indicator line on the zero bar will be calculated. When the cycle is finished, start() will stop executing and will pass control to the client terminal.

Variant 2. A new tick is the first tick of a zero bar (happens from time to time).

Fig. 118. The processed tick is the first tick of a new zero bar.

In this case the fact of appearance of a new bar is important. Before control is passed to the special function start(), client terminal will draw again all bars present in the security window and re-index all declared indicator arrays (set in correspondence with buffers). Besides, client terminal will remember that there are already 301 bars, not 300 in a chart window.

Fig. 118 contains situation when on the last tick of the previous bar (at the moment t 2) the function start() was successfully started and executed. That's why, though now the first bar (with index 1) finished at the moment t 2 was calculated by the indicator, function IndicatorCounted() will return value that was on the previous bar, i.e. 299: 

   Counted_bars=IndicatorCounted(); // Number of counted bars
In the next line index i will be calculated, in this case for the first tick of a new bar it will be equal to 1 (301-299-1):

   i=Bars-Counted_bars-1;           // Index of the first uncounted
It means calculation of indicator array values in while() loop at the appearance of a new bar will be performed both for the last bar and for the new zero bar. A little earlier during re-indexation of indicator arrays the client terminal increased sizes of these arrays. Values of array elements with zero indexes were not defined before the calculations in the loop. During calculations in the loop these elements get some values. When calculations in start() are over, control is returned to the client terminal. After that the client terminal will draw indicator lines on the zero bar based on just calculated values of array elements with zero indexes.

Variant 3. A new tick is the first tick of a new zero bar, but the last but one tick is not processed (rare case).

Fig. 119. Not all ticks of the previous bar were processed.

Fig. 119 shows the situation when start() was launched on the first tick of a new bar at the moment t 5. Previous time this function was started at the moment t 2. Tick that came to the terminal at the moment t 3 (red arrow) was not processed by the indicator. This happened because start() execution time t 2 - t 4 is larger than the interval between ticks t 2 - t 3. This fact will be detected by the client terminal during the execution of start() launched at the moment t 5. During calculations in the line:

   Counted_bars=IndicatorCounted(); // Number of counted bars
IndicatorCounted() will return the value 299 (!). This value is true - from the moment of the last indicator call 299 bars were not changed after (now already) 301. That is why the calculated index of the first (leftmost) bar, from which calculations of array element values must be started, will be equal to 1 (301-299-1):

   i=Bars-Counted_bars-1;           // Index of the first uncounted
it means during while() execution two iterations will be performed. During the first one values of array elements with the index i = 1 will be calculated, i.e. Buf_0[1] and Buf_1[1]. Not, by the moment calculations start, bars and indicator arrays are already re-indexed by the client terminal (because a new bar started, between starts of the special function start()). That is why calculations for elements of arrays with index 1 will be calculated on the basis of array-timeseries (maximal and minimal values of a bar price) also with the index 1:

   while(i>=0)                      // Loop for uncounted bars
     {
      Buf_0[i] = High[i];           // Value of 0 buffer on i bar
      Buf_1[i] = Low[i];            // Value of 1st buffer on i bar
      i--;                          // Calculating index of the next bar
     }
During the second iteration of while() values for elements with zero indexes, i.e. for the zero bar, is calculated on the basis of last known values of arrays-timeseries.

	Using of the described technology for the calculation of custom indicators allows, first, to guarantee calculation of values of all indicator array elements irrespective of the specific nature of tick history, and second, to conduct calculations only for uncounted bars, i.e. use economically calculating resources.


Not, a bar is considered uncounted if calculation of element values of an indicator arrays at least for one last tick of the bar is not performed. 

Starting the custom indicator userindicator.mq4 in a chart window you will see two lines - a thick blue line built upon bar maximums and a dotted red line built upon its minimums (Fig. 120).

Fig. 120. Two indicator lines in a security window, built by the indicator userindicator.mq4.

It should be noted, that one can built a custom indicator, indicator lines of which would coincide with the lines of an analogous technical indicator. It can be easily done if as calculation formulas in the custom indicator, the same formulas as in the technical indicator are used. To illustrate this let's improve the program code analyzed in the previous example. Let the indicator draw lines upon average values of maximums and minimums of several last bars. It is easy to conduct necessary calculations: we simply need to find average values of arrays-timeseries elements. For example, value of an indicator array with the index 3 (i.e. indicator line coordinate for the third bar) on the basis of the last five maximums is calculated the following way:

Buf_0[3] = ( High[3] + High[4] + High[5] + High[6] + High[7] ) / 5

Analogous calculations can be performed for an indicator lines built upon minimums.

	Example of a simple custom indicator averagevalue.mq4. Indicator lines are built upon average minimal and maximal values of N bars.


//--------------------------------------------------------------------
// averagevalue.mq4 
// The code should be used for educational purpose only.
//--------------------------------------------------------------------
#property indicator_chart_window    // Indicator is drawn in the main window
#property indicator_buffers 2       // Number of buffers
#property indicator_color1 Blue     // Color of the 1st line
#property indicator_color2 Red      // Color of the 2nd line
 
extern int Aver_Bars=5;             // number of bars for calculation
 
double Buf_0[],Buf_1[];             // Declaring indicator arrays
//--------------------------------------------------------------------
int init()                          // Special function init()
  {
//--------------------------------------------------------------------
   SetIndexBuffer(0,Buf_0);         // Assigning an array to a buffer
   SetIndexStyle (0,DRAW_LINE,STYLE_SOLID,2);// Line style
//--------------------------------------------------------------------
   SetIndexBuffer(1,Buf_1);         // Assigning an array to a buffer
   SetIndexStyle (1,DRAW_LINE,STYLE_DOT,1);// Line style
//--------------------------------------------------------------------
   return;                          // Exit the special funct.init()
  }
//--------------------------------------------------------------------
int start()                         // Special function start()
  {
   int i,                           // Bar index
       n,                           // Formal parameter
       Counted_bars;                // Number of counted bars
       double
       Sum_H,                       // Sum of High values for period
       Sum_L;                       // Sum of Low values for period
//--------------------------------------------------------------------
   Counted_bars=IndicatorCounted(); // Number of counted bars
   i=Bars-Counted_bars-1;           // Index of the first uncounted
   while(i>=0)                      // Loop for uncounted bars
     {
      Sum_H=0;                      // Nulling at loop beginning
      Sum_L=0;                      // Nulling at loop beginning
      for(n=i;n<=i+Aver_Bars-1;n++) // Loop of summing values
        {
         Sum_H=Sum_H + High[n];     // Accumulating maximal values sum
         Sum_L=Sum_L + Low[n];      // Accumulating minimal values sum
        }
      Buf_0[i]=Sum_H/Aver_Bars;     // Value of 0 buffer on i bar
      Buf_1[i]=Sum_L/Aver_Bars;     // Value of 1st buffer on i bar
 
      i--;                          // Calculating index of the next bar
     }
//--------------------------------------------------------------------
   return;                          // Exit the special funct. start()
  }
//--------------------------------------------------------------------
In this example there is an external variable Aver_Bars. Using this variable a user can indicate the number of bars, for which an average value is calculated. In start()this value is used for the calculation of an average value. In the loop 'for' the sum of maximal and minimal values is calculated for the number of bars corresponding to the value of the variable Aver_Bars. In the next two program lines values of indicator array elements are calculated for indicator lines corresponding to minimal and maximal values.

The averaging method used here is also applied for calculations in the technical indicator Moving Average. If we attach the analyzed custom indicator averagevalue.mq4 and the technical indicator Moving Average, we will see three indicator lines. If the same period of averaging is set up for both indicators, Moving Average line will coincide with one of the custom indicator lines (for this purpose parameters described in Fig. 121 must be specified in the technical indicator settings).

Fig. 121. Coincident lines of a technical indicator and a custom indicator (red line).

Thus, using technical indicator a user can construct the reflection of any regularities necessary in practical work.

Custom Indicator Options

Drawing Indicator Lines in Separate Windows

MQL4 offers a large service for constructing custom indicators which makes using them very convenient. In particular, indicator lines can be drawn in a separate window. This is convenient when absolute values of the indicator line amplitude is substantially smaller (or larger) than security prices. For example, if we are interested in the difference between average values of bar maximums and minimums in a certain historic interval, depending on timeframe this value will be equal to approximately from 0 to 50 points (for example, for M15). It is not difficult to build an indicator line, but in a security window this line will be drawn in the range 0 - 50 points of a security price, i.e. substantially lower than the chart area reflected on the screen. It is very inconvenient.

To draw indicator lines in a separate window (which is in the lower part of a security window), in the directive #property (at the program beginning) parameter indicator_separate_window must be specified:

#property indicator_separate_window // Indicator is drawn in a separate window
At the moment when such an indicator is attached to a security window, client terminal creates a separate window below a chart, in which indicator lines calculated in the indicator will be drawn. Depending on color settings and types of indicator lines they will be drawn in this or that style.

Limiting Calculation History

In most cases indicator lines contain useful information only in the most recent history. The part of indicator lines built upon old bars (for example, 1 month old minute timeframe) can hardly be considered useful for making trade decisions. Besides, if there are a lot of bars in a chart window, time invested into the calculation and drawing of indicator lines is unreasonably large. This may be critical in program debugging, when a program is often compiled and then started. That is why it is necessary to conduct calculations not for the whole history, but for the limited part of the most recent bar history.

For this purpose an external variable history is used in the following program. Value of this variable is taken into account when calculating index of the first (leftmost) bar, starting from which elements of indicator arrays must be calculated.

   i=Bars-Counted_bars-1;           // Index of the first uncounted
   if (i>History-1)                 // If there are too many bars...
      i=History-1;                  // ..calculate for specified amount.
Further calculations in while() loop will be conducted for the number of recent history bars not larger than History value. Note, the analyzed method of limiting a calculation history concerns only the part of calculations that are conducted in the first start of the special function start(). Further, when new bars appear, new parts of indicator lines will be added in the right part, while the image in the left part will be preserved. Thus the indicator line length will be increased during the whole indicator operation time. Common value of History parameter is considered approximately 5000 bars.

	Example of a simple custom indicator separatewindow.mq4. Indicator lines are drawn in a separate window.


//--------------------------------------------------------------------
// separatewindow.mq4 
// The code should be used for educational purpose only.
//--------------------------------------------------------------------
#property indicator_separate_window // Drawing in a separate window
#property indicator_buffers 1       // Number of buffers
#property indicator_color1 Blue     // Color of the 1st line
#property indicator_color2 Red      // Color of the 2nd line
 
extern int History  =50;            // Amount of bars in calculation history
extern int Aver_Bars=5;             // Amount of bars for calculation
 
double Buf_0[];                     // Declaring an indicator array
//--------------------------------------------------------------------
int init()                          // Special function init()
  {
   SetIndexBuffer(0,Buf_0);         // Assigning an array to a buffer
   SetIndexStyle (0,DRAW_LINE,STYLE_SOLID,2);// line style
   return;                          // Exit the special funct. init()
  }
//--------------------------------------------------------------------
int start()                         // Special function start()
  {
   int i,                           // Bar index
   n,                               // Formal parameter
   Counted_bars;                    // Number of counted bars
   double
   Sum_H,                           // Sim of High values for period
   Sum_L;                           // Sum of low values for period
//--------------------------------------------------------------------
   Counted_bars=IndicatorCounted(); // Number of counted bars
   i=Bars-Counted_bars-1;           // Index of the first uncounted
   if (i>History-1)                 // If too many bars ..
      i=History-1;                  // ..calculate for specific amount.
   while(i>=0)                      // Loop for uncounted bars
     {
      Sum_H=0;                      // Nulling at loop beginning
      Sum_L=0;                      // Nulling at loop beginning
      for(n=i;n<=i+Aver_Bars-1;n++) // Loop of summing values 
        {
         Sum_H=Sum_H + High[n];     // Accumulating maximal values sum
         Sum_L=Sum_L + Low[n];      // Accumulating minimal values sum
        }
      Buf_0[i]=(Sum_H-Sum_L)/Aver_Bars;// Value of 0 buffer on i bar
      i--;                          // Calculating index of the next bar
     }
//--------------------------------------------------------------------
   return;                          // Exit the special funct. start()
  }
//--------------------------------------------------------------------
Similar calculation of an indicator line is performed in the technical indicator AverageTrue Range. Fig. 122 shows an indicator line constructed by the custom indicator separatewindow.mq4 in a separate window and an indicator line constructed by ATR in another window. In this case lines are fully identical because period of averaging is the same for both indicators - 5. If this parameter is changed in any of the indicators, the corresponding indicator line will also change. 

Fig. 122. drawing a custom indicator line in a separate window.
Identical lines of a technical indicator (ATR) and a custom indicator (separatewindow.mq4).

It is also evident that custom indicator line is constructed not for the whole screen width, but for 50 latest bars as specified in the external variable History. If a trader needs to use larger history interval, value of the external variable can be easily changed via the custom indicator settings window.

Fig. 123 shows a security window, in which the indicator line us drawn in another style - as a histogram. For getting such a result one line was changed in the program code separatewindow.mq4 - other line styles are indicated:

   SetIndexStyle (0,DRAW_HISTOGRAM);// Line style
All other code parts are unchanged.

Fig. 123. Drawing custom indicator line in a separate window (histogram).
Similarity of drawings of a technical indicator (ATR) and a custom indicator (separatewindow.mq4).

Shifting Indicator Lines Vertically and Horizontally

In some cases it is necessary to shift an indicator line. It can be easily done by MQL4 means. Let's analyze an example, in which position of indicator lines in a security window are calculated in accordance with values specified by a user.

	Example of a custom indicator displacement.mq4. Shifting indicator lines horizontally and vertically.


//--------------------------------------------------------------------
// displacement.mq4 
// The code should be used for educational purpose only.
//--------------------------------------------------------------------
#property indicator_chart_window       //Indicator is drawn in the main window
#property indicator_buffers 3          // Number of buffers
#property indicator_color1 Red         // Color of the 1st line
#property indicator_color2 Blue        // Color of the 2nd line
#property indicator_color3 Green       // Color of the 3rd line
 
extern int History  =500;              // Amount of bars in calculation history
extern int Aver_Bars=5;                // Amount of bars for calculation
extern int Left_Right= 5;              // Horizontal shift (bars)
extern int Up_Down  =25;               // Vertical shift (points)
 
double Line_0[],Line_1[],Line_2[];     // Declaring data arrays
//--------------------------------------------------------------------
int init()                             // Special funct. init()
  {
//--------------------------------------------------------------------
   SetIndexBuffer(0,Line_0);           // Assigning an array to buffer 0
   SetIndexStyle (0,DRAW_LINE,STYLE_SOLID,2);// Line style
//--------------------------------------------------------------------
   SetIndexBuffer(1,Line_1);           // Assigning an array to buffer 1
   SetIndexStyle (1,DRAW_LINE,STYLE_DOT,1);// Line style
//--------------------------------------------------------------------
   SetIndexBuffer(2,Line_2);           // Assigning an array to buffer 2
   SetIndexStyle (2,DRAW_LINE,STYLE_DOT,1);// Line style
//--------------------------------------------------------------------
   return;                             // Exit the special funct. init()
  }
//--------------------------------------------------------------------
int start()                            // Special function start()
  {
   int i,                              // Bar index
       n,                              // Formal parameter (index)
       k,                              // Index of indicator array element
       Counted_bars;                   // Number of counted bars
       double
       Sum;                            // High and Low sum for the period
//--------------------------------------------------------------------
   Counted_bars=IndicatorCounted();    // Number of counted bars
   i=Bars-Counted_bars-1;              // Index of the 1st uncounted
   if (i>History-1)                    // If too many bars ..
      i=History-1;                     // ..calculate for specified amount.
 
   while(i>=0)                         // Loop for uncounted bars
     {
      Sum=0;                           // Nulling at loop beginning
      for(n=i;n<=i+Aver_Bars-1;n++)    // Loop of summing values 
         Sum=Sum + High[n]+Low[n];     // Accumulating maximal values sum
      k=i+Left_Right;                  // Obtaining calculation index
      Line_0[k]= Sum/2/Aver_Bars;      // Value of 0 buffer on k bar
      Line_1[k]= Line_0[k]+Up_Down*Point;// Value of the 1st buffer
      Line_2[k]= Line_0[k]-Up_Down*Point;// Value of the 2nd buffer
 
      i--;                             // Calculating index of the next bar
     }
//--------------------------------------------------------------------
   return;                             // Exit the special funct. start()
  }
//--------------------------------------------------------------------
For adjusting lines shift in a chart, there are two external variables - Left_Right for horizontal shift of all lines and Up_Down for shifting two dotted lines vertically.

extern int Left_Right= 5;              // Horizontal shift (bars)
extern int Up_Down   = 25;             // Vertical shift (points)
The algorithm used for calculating values of corresponding array elements is based on very simple rules:

· for shifting a line horizontally, assign the calculated value to an array element, the index of which is larger by Left_Right (for shifting to the right and less for shifting to the right) than the index of a bar, for which calculations are conducted; 

· for shifting a line vertically, Up_Down*Point must be added (for shifting upwards or detracted for shifting downwards) to each value of an indicator array characterizing initial line position; 

In the analyzed example indexes are calculated in the line:

      k = i+Left_Right;                // Obtaining calculation index
Here i is the index of a bar, for which calculations are performed, k is an index of an indicator array element. Red indicator line displayed by the client terminal based on the indicator array Line_0[] is shifted to the left by 5 bars (according to custom settings, see Fig. 124) from the initial line. In this case the initial line is a Moving Average with the period of averaging equal to 5; the formula of MA calculation is (High[i]+Low[i])/2 .

      Line_0[k]= Sum2 Aver_Bars;       // Value of 0 buffer on k bar
In this example the position of the red line is the basis for the calculation of indicator array values for two other lines, i.e. their position on the chart. Dotted lines are calculated this way:

      Line_1[k]= Line_0[k]+Up_Down*Point;// Value of the 1st buffer 
      Line_2[k]= Line_0[k]-Up_Down*Point;// Value of the 2nd buffer
Use of index k for elements of all indicator arrays allows to perform calculations for elements of arrays Line_1[], Line_2[] on the same bar as used for calculating values of the corresponding basic array Line_0[]. As a result dotted lines are shifted relative to the red line by the value specified in the indicator settings window, in this case by 30 points (Fig. 124).

Fig. 124. Red indicator line is shifted to the left by 5 bars.
Dotted indicator lines are shifted relative to the red line by 30 points.

Limitations of Custom Indicators

There are some limitations in MQL4 that should be taken into account in the programming of custom indicators.

There is a group of functions that can be used only in custom indicators and cannot be used in Expert Advisors and scripts: IndicatorBuffers(), IndicatorCounted (), IndicatorDigits(), IndicatorShortName(), SetIndexArrow(), SetIndexBuffer(), SetIndexDrawBegin(), SetIndexEmptyValue(), SetIndexLabel(), SetIndexShift(), SetIndexStyle(), SetLevelStyle(), SetLevelValue().

On the other hand, trade functions cannot be used in indicators: OrderSend(), OrderClose(), OrderCloseBy(), OrderDelete() and OrderModify(). This is because indicators operate in the interface flow (as distinct from Expert Advisors and scripts that operate in their own flow).

This is also why algorithms based on looping cannot be used in custom indicators. Start of a custom indicator containing an endless loop (in terms of actual execution time) can result in client terminal hanging up with further necessity to restart a computer.

The general comparative characteristics of Expert Advisors, scripts and indicators is contained in Table 2.

Custom Indicator ROC (Price Rate of Change)

It is known, all indicators are of application relevance - they are used to help a trader orientate in the current price movement and forecast at least to some extent the future price movement. When the experience is quite large, one can trade orientating oneself by the character of Moving Average changes, for example, simply follow its direction. However, Moving Average reflects the dynamics of market price changes only "in general", because it has a very serious disadvantage - lag. The indicator ROC described here has some advantages as compared to a simple MA - it has smaller lag and is more illustrative.

Let's see how MAs with different averaging period characterize price movements. Fig. 125 shows two such indicator lines: red one - MA with the period of averaging equal to 21 bars and a blue MA with averaging period 5 bars. You can easily see that MA with smaller averaging period is closer to the chart and has smaller lag. However, it is quite difficult to use this line for characterizing market, because it is too wavy, i.e. very often changes its direction, thus giving a lot of false signals. MA with a larger averaging period is not so wavy, i.e. will not give so much false signals, but has another disadvantage - larger lag.

Fig. 125. Indicator lines: MA(21) - red, MA(5) - blue, ROC - orange. 

The third line present in Fig. 125 is an indicator line of rate of change (orange). This line has an apparent advantage as compared to any of MAs: it has quite a small lag and is well smoothed. Let's discuss the line in details.

This indicator line is built on the basis of the rate of MA(21) change. In part A-B rate of MA change grows. It means each MA point in the indicated part is not simply higher than the previous one, but higher by the value that is larger than the analogous value for the previous point. For example, if on the bar with index 271 MA(21) value was 1.3274, on the bar with index 272 - 1.3280, on bar 273 - 1.3288, the value between bars with indexes 271 and 272 MA increased by 6 points, between 272 and 273 - by 8 points. Thus MA not simply grows, but its rate of change also increases. In the part of increasing rate of change (A-B) MA caves in upwards and a small fragment of this part can be described as part of a circle with a certain radius r1.

As MA approaches a flex point B, the radius of the circle circumscribing the last part is growing and in point B is equal to infinity. I.e. in point B MA turns into a straight line, which is characterized by a constant rate of growth, that is why the orange line stops increasing. In the part B-C MA's growing slows down, but goes on. Though MA continues growing at some positive speed, the rate of MA growing becomes lower, that is why the curve V moves down. Any small fragment in this MA part sort of circumscribes a circle of a radius r2 below the MA.

In point C MA stops growing, i.e. its speed is equal to zero. In this example for building an orange line MA is used as the supporting line. Here the notion of supporting MA should be specified. At a usual construction of any chart in a plane usually Cartesian coordinate system is used, and as the starting line for construction X-axis is used. In our case as such a line not a straight axis is used, but MA with a certain period of averaging (in this case MA(21), red line), it is called a supporting MA. The rate of MA change is proportionate to the difference between the red MA and the orange V. I.e. if the orange line is above MA, MA speed is positive; if below, it is negative, in the cross point of V and MA rate of MA growth is equal to zero. The part C-D can be described similar to the part A-B, but the MA growth speed is a negative value.

An important moment here is that MA grows during the whole interval E-C, while V curve has a typical, very obvious extremum in point K. Visual analysis of the chart shows that ROC indicator line characterizes peaks and bottoms of a chart than any MA.

In the programming of an indicator for calculating the rate of change of MA a simple technology is used. Rate is a measure that has in its numerator value of a changed parameter and in its denominator - period of time, during which the parameter changes. In the context of this indicator (see Fig. 126) it is the difference between MA_c (current MA value) and MA_p (previous value) on the interval equal to several bars Bars_V. Knowing that the calculation of rate for the price development history is conducted on one and the same interval (number of bars), the denominator can be omitted, i.e. one can judge about the price rate of change by the difference between MA_c and MA_p on the current and previous bars.

Fig. 126. Parameters for constructing ROC indicator line.

The analyzed custom indicator calculates 6 indicator lines in all. The indicator array Line_0[] contains values of the supporting MA, relative to which all other indicator lines are constructed. Next three indicator arrays (Line_1[], Line_2[] and Line_3[]) contain values of the rates of price changes based on MAs with different periods of averaging. The indicator array Line_4[] is intended for building an average rate line (arithmetic average of Line_1[],Line_2[] and Line_3[]), and Line_5[] - for constructing the same rate average line, but smoothed one.

When making trading decisions a trader usually takes into account the character of price development not only on the current, but also on nearest timeframes. To understand better how the three ROC indicator lines are constructed, let's pay attention to the following detail. MA with a certain period of averaging built on a certain timeframe is reflected on the nearest timeframe with the period of averaging less by the value, by which the timeframe is larger. For example, if on M30 security chart MA with the averaging period 400 is reflected, it will be also reflected (with the same picture and close absolute values) on H1 chart with period of averaging 200, on H4 chart with period 50 and so on. Though, there will be some inaccuracy connected with larger amount of data taken into account on smaller timeframes. However, in most cases this inaccuracy is acceptably small.

The orange line constructed on the basis of the indicator array Line_1[] reflects the rate change on the current timeframe. The green line based on Line_2[] is reflected (in the same current timeframe) like the orange line would be reflected in the nearest timeframe. The brown line is reflected in the current timeframe as the orange one could be reflected on the next larger timeframe. Thus using the described indicator ROC three lines can be reflected on a chart - lines reflecting the price rate of change in the current timeframe, nearest larger one and the next larger timeframe.

	Custom indicator roc.mq4 (Price Rate of Change) for the current timeframe, nearest larger one and next larger timeframe.


//--------------------------------------------------------------------
// roc.mq4 (Priliv)
// The code should be used for educational purpose only.
//--------------------------------------------------------------------
//--------------------------------------------------------------- 1 --
#property copyright "Copyright © SK, 2007"
#property link      "http://AutoGraf.dp.ua"
//--------------------------------------------------------------------
#property indicator_chart_window    // Indicator is drawn in the main window
#property indicator_buffers 6       // Number of buffers
#property indicator_color1 Black    // Line color of 0 buffer
#property indicator_color2 DarkOrange//Line color of the 1st buffer
#property indicator_color3 Green    // Line color of the 2nd buffer
#property indicator_color4 Brown    // Line color of the 3rd buffer
#property indicator_color5 Blue     // Line color of the 4th buffer
#property indicator_color6 Red      // Line color of the 5th buffer
//--------------------------------------------------------------- 2 --
extern int History    =5000;        // Amount of bars for calculation history
extern int Period_MA_0=13;          // Period of supporting MA for cur. timefr.
extern int Period_MA_1=21;          // Period of calculated MA
extern int Bars_V     =13;          // Amount of bars for calc. rate
extern int Aver_Bars  =5;           // Amount of bars for smoothing
extern double K       =2;           // Amplifier gain
//--------------------------------------------------------------- 3 --
int
   Period_MA_2,  Period_MA_3,       // Calculation periods of MA for other timefr.
   Period_MA_02, Period_MA_03,      // Calculation periods of supp. MAs
   K2, K3;                          // Coefficients of timeframe correlation
double
   Line_0[],                        // Indicator array of supp. MA
   Line_1[], Line_2[], Line_3[],    // Indicator array of rate lines
   Line_4[],                        // Indicator array - sum
   Line_5[],                        // Indicator array - sum, smoothed
   Sh_1, Sh_2, Sh_3;                // Amount of bars for rates calc.
//--------------------------------------------------------------- 4 --
int init()                          // Special function init()
  {
   SetIndexBuffer(0,Line_0);        // Assigning an array to a buffer
   SetIndexBuffer(1,Line_1);        // Assigning an array to a buffer
   SetIndexBuffer(2,Line_2);        // Assigning an array to a buffer
   SetIndexBuffer(3,Line_3);        // Assigning an array to a buffer
   SetIndexBuffer(4,Line_4);        // Assigning an array to a buffer
   SetIndexBuffer(5,Line_5);        // Assigning an array to a buffer
   SetIndexStyle (5,DRAW_LINE,STYLE_SOLID,3);// line style
//--------------------------------------------------------------- 5 --
   switch(Period())                 // Calculating coefficient for..
     {                              // .. different timeframes
      case     1: K2=5;K3=15; break;// Timeframe M1
      case     5: K2=3;K3= 6; break;// Timeframe M5
      case    15: K2=2;K3= 4; break;// Timeframe M15
      case    30: K2=2;K3= 8; break;// Timeframe M30
      case    60: K2=4;K3=24; break;// Timeframe H1
      case   240: K2=6;K3=42; break;// Timeframe H4
      case  1440: K2=7;K3=30; break;// Timeframe D1
      case 10080: K2=4;K3=12; break;// Timeframe W1
      case 43200: K2=3;K3=12; break;// Timeframe MN
     }
//--------------------------------------------------------------- 6 --
   Sh_1=Bars_V;                     // Period of rate calcul. (bars)
   Sh_2=K2*Sh_1;                    // Calc. period for nearest TF
   Sh_3=K3*Sh_1;                    // Calc. period for next TF
   Period_MA_2 =K2*Period_MA_1;     // Calc. period of MA for nearest TF
   Period_MA_3 =K3*Period_MA_1;     // Calc. period of MA for next TF
   Period_MA_02=K2*Period_MA_0;     // Period of supp. MA for nearest TF
   Period_MA_03=K3*Period_MA_0;     // Period of supp. MA for next TF
//--------------------------------------------------------------- 7 --
   return;                          // Exit the special function init()
  }
//--------------------------------------------------------------- 8 --
int start()                         // Special function start()
  {
//--------------------------------------------------------------- 9 --
   double
   MA_0, MA_02, MA_03,              // Supporting MAs for diff. TF
   MA_c, MA_p,                      // Current and previous MA values
   Sum;                             // Technical param. for sum accumul.
   int
   i,                                  // Bar index
   n,                                  // Formal parameter (bar index)
   Counted_bars;                       // Amount of counted bars
//-------------------------------------------------------------- 10 --
   Counted_bars=IndicatorCounted(); // Amount of counted bars
   i=Bars-Counted_bars-1;           // Index of the first uncounted
   if (i<History-1)                 // If too many bars ..
      i=History-1;                  // ..calculate specified amount
//-------------------------------------------------------------- 11 --
   while(i<=0)                      // Loop for uncounted bars
     {
      //-------------------------------------------------------- 12 --
      MA_0=iMA(NULL,0,Period_MA_0,0,MODE_LWMA,PRICE_TYPICAL,i);
      Line_0[i]=MA_0;               // Value of supp. MA
      //-------------------------------------------------------- 13 --
      MA_c=iMA(NULL,0,Period_MA_1,0,MODE_LWMA,PRICE_TYPICAL,i);
      MA_p=iMA(NULL,0,Period_MA_1,0,MODE_LWMA,PRICE_TYPICAL,i+Sh_1);
      Line_1[i]= MA_0+K*(MA_c-MA_p);// Value of 1st rate line
      //-------------------------------------------------------- 14 --
      MA_c=iMA(NULL,0,Period_MA_2,0,MODE_LWMA,PRICE_TYPICAL,i);
      MA_p=iMA(NULL,0,Period_MA_2,0,MODE_LWMA,PRICE_TYPICAL,i+Sh_2);
      MA_02= iMA(NULL,0,Period_MA_02,0,MODE_LWMA,PRICE_TYPICAL,i);
      Line_2[i]=MA_02+K*(MA_c-MA_p);// Value of 2nd rate line
      //-------------------------------------------------------- 15 --
      MA_c=iMA(NULL,0,Period_MA_3,0,MODE_LWMA,PRICE_TYPICAL,i);
      MA_p=iMA(NULL,0,Period_MA_3,0,MODE_LWMA,PRICE_TYPICAL,i+Sh_3);
      MA_03= iMA(NULL,0,Period_MA_03,0,MODE_LWMA,PRICE_TYPICAL,i);
      Line_3[i]=MA_03+K*(MA_c-MA_p);// Value of 3rd rate line
      //-------------------------------------------------------- 16 --
      Line_4[i]=(Line_1[i]+Line_2[i]+Line_3[i])/3;// Summary array
      //-------------------------------------------------------- 17 --
      if (Aver_Bars>0)              // If wrong set smoothing
         Aver_Bars=0;               // .. no less than zero
      Sum=0;                        // Technical means
      for(n=i; n>=i+Aver_Bars; n++) // Summing last values
         Sum=Sum + Line_4[n];       // Accum. sum of last values
      Line_5[i]= Sum/(Aver_Bars+1); // Indic. array of smoothed line
      //-------------------------------------------------------- 18 --
      i--;                          // Calculating index of the next bar
      //-------------------------------------------------------- 19 --
     }
   return;                          // Exit the special function start()
  }
//-------------------------------------------------------------- 20 --
To calculate indicator arrays of three rate lines MAs with different averaging periods are used. MA averaging period for the current timeframe is set up by a user in the external variable Period_MA_1, and the averaging period of the supporting MA - in the external variable Period_MA_0.

Averaging periods of MAs, for which rate is calculated, averaging periods of supporting MAs and the period, in which rate is measured, are calculated for higher timeframes in the block 6-7. Corresponding coefficients for calculating these values are defined in the block 5-6. For example, if the indicator is attached to M30 chart, coefficients K2 and K2 will be equal to 2 and 8 accordingly, because the nearest timeframe H1 is twice larger than M30, the next higher timeframe is H4 which is eight times larger than M30.

Calculations in start() are very simple. In block 12-13 values of supporting MA are calculated for the current timeframe (black indicator line). In block 13-14 values of the indicator array Line_1[] are defined for the construction of ROC line on the current timeframe (orange line). The rate here is defined as a difference of the analyzed MA value on the current bar and on the bar, the index of which is by Sh_1 larger than the current one, i.e. (MA_c - MA_p). The value of the indicator array Line_1[] on the current bar is made up of values of the supporting MA and a value characterizing rate (here K is a scale coefficient set up in an external variable):

      Line_1[i]= MA_0+K*(MA_c-MA_p);// value of 1st rate line
Analogous calculations are conducted for constructing rate lines for two other timeframes (blocks 14-16). Supporting MAs for these arrays are not shown by the indicator. In the block 16017 values of the indicator array Line_4[] are defined for constructing an average rate line (blue line), which is their simple arithmetic average.

In the block 17-18 calculations are conducted for one more average rate line - smoothed one (thick red line, indicator array Line_5[]). Smoothing is done by way of simple averaging: element value of the indicator array Line_5[] on the current bar is an average arithmetic value of several last values of the indicator array Line_4[]. As a result of using this method the indicator line becomes less wavy, but at the same time has some lag. Amount of bars for smoothing is set in the external variable Aver_Bars.

Starting the indicator you will see 6 indicator lines in a chart window:

· black line - supporting MA for building a price rate line on the current timeframe; 

· orange line - price rate of change on the current timeframe; 

· green line - price rate of change on the nearest higher timeframe; 

· brown line - price rate of change on the next higher timeframe; 

· blue line - average line of the rate of price change;

· red line - smoothed average line of the rate of price change. 

Fig. 127. Custom indicator roc.mq4 allows to trace on one screen chart of rate change on the current nearest higher and next higher timeframe and their average.

Indicator roc.mq4 can be attached to the window of any security with any timeframe. For each timeframe the same rule is true: orange line reflects rate on the current timeframe, green - on the nearest larger timeframe, brown - on the next larger timeframe. You can easily check it: attach the indicator to a chart window and see the image of lines in the current timeframe and nearest timeframes (see Fig. 128 and Fig. 129).

Fig. 128. Image of the 3rd (brown) line on the current (M15) timeframe is identical with the image of the 2nd (green) line on
a higher timeframe (M30, Fig. 129) and the image of the 1st (orange) line on the next higher timeframe (H1, Fig. 129).

Fig. 129. Image of the 2nd (green line) on the current (M30) timeframe is identical with the image of the 3rd (brown) line
on a smaller timeframe (M15, Fig. 128) and the image of the 1st (orange) line on a higher timeframe (H1).

There is one peculiarity in the analyzed indicator roc.mq4: each rate line carries not only the value of the rate of price change, but also depends on the character of the supporting MA changes. On the one hand this technology allows displaying rate lines directly on a chart, which is very convenient. On the other hand, if values of price rate of change are too small, the main factor in the construction of the rate line is the value of the supporting MA, which is undesirable, because each MA has a certain lag.

The next custom indicator is the full analogue of the indicator roc.mq4, but it is drawn in a separate window. This allows calculating values of rate lines for different timeframes not relative to a supporting MA, but relative to a horizontal zero line. Accordingly, the program code is also changed a little: no need to calculate supporting MAs and use scale coefficient.

	Custom indicator rocseparate.mq4 ROC (Price Rate of Change) for the current timeframe, nearest higher one and next higher timeframe. Displayed in a separate window.


//--------------------------------------------------------------------
// rocseparate.mq4 (Priliv_s)
// The code should be used for educational purpose only.
//--------------------------------------------------------------- 1 --
#property copyright "Copyright © SK, 2007"
#property link      "http://AutoGraf.dp.ua"
//--------------------------------------------------------------------
#property indicator_separate_window // Indicator is drawn in a separate window
#property indicator_buffers 6       // Number of buffers
#property indicator_color1 Black    // Line color of 0 buffer
#property indicator_color2 DarkOrange//Line color of the 1st buffer
#property indicator_color3 Green    // Line color of the 2nd buffer
#property indicator_color4 Brown    // Line color of the 3rd buffer
#property indicator_color5 Blue     // Line color of the 4th buffer
#property indicator_color6 Red      // Line color of the 5th buffer
//--------------------------------------------------------------- 2 --
extern int History    =5000;        // Amount of bars in calculation history
extern int Period_MA_1=21;          // Period of calculated MA
extern int Bars_V     =13;          // Amount of bars for calc. rate
extern int Aver_Bars  =5;           // Amount of bars for smoothing
//--------------------------------------------------------------- 3 --
int
   Period_MA_2,  Period_MA_3,       // Calculation periods of MA for other timefr.
   K2, K3;                          // Coefficients of timeframe correlation
double
   Line_0[],                        // Indicator array of supp. MA
   Line_1[], Line_2[], Line_3[],    // Indicator array of rate lines 
   Line_4[],                        // Indicator array - sum
   Line_5[],                        // Indicator array - sum, smoothed
   Sh_1, Sh_2, Sh_3;                // Amount of bars for rates calc.
//--------------------------------------------------------------- 4 --
int init()                          // Special function init()
  {
   SetIndexBuffer(0,Line_0);        // Assigning an array to a buffer
   SetIndexBuffer(1,Line_1);        // Assigning an array to a buffer
   SetIndexBuffer(2,Line_2);        // Assigning an array to a buffer
   SetIndexBuffer(3,Line_3);        // Assigning an array to a buffer
   SetIndexBuffer(4,Line_4);        // Assigning an array to a buffer
   SetIndexBuffer(5,Line_5);        // Assigning an array to a buffer
   SetIndexStyle (5,DRAW_LINE,STYLE_SOLID,3);// Line style
//--------------------------------------------------------------- 5 --
   switch(Period())                 // Calculating coefficient for..
     {                              // .. different timeframes
      case     1: K2=5;K3=15; break;// Timeframe M1
      case     5: K2=3;K3= 6; break;// Timeframe M5
      case    15: K2=2;K3= 4; break;// Timeframe M15
      case    30: K2=2;K3= 8; break;// Timeframe M30
      case    60: K2=4;K3=24; break;// Timeframe H1
      case   240: K2=6;K3=42; break;// Timeframe H4
      case  1440: K2=7;K3=30; break;// Timeframe D1
      case 10080: K2=4;K3=12; break;// Timeframe W1
      case 43200: K2=3;K3=12; break;// Timeframe MN
     }
//--------------------------------------------------------------- 6 --
   Sh_1=Bars_V;                     // Period of rate calcul. (bars)
   Sh_2=K2*Sh_1;                    // Calc. period for nearest TF
   Sh_3=K3*Sh_1;                    // Calc. period for next TF
   Period_MA_2 =K2*Period_MA_1;     // Calc. period of MA for nearest TF
   Period_MA_3 =K3*Period_MA_1;     // Calc. period of MA for next TF
//--------------------------------------------------------------- 7 --
   return;                          // Exit the special function init()
  }
//--------------------------------------------------------------- 8 --
int start()                         // Special function start()
  {
//--------------------------------------------------------------- 9 --
   double
   MA_c, MA_p,                      // Current and previous MA values
   Sum;                             // Technical param. for sum accumul.
   int
   i,                               // Bar index
   n,                               // Formal parameter (bar index)
   Counted_bars;                    // Amount of counted bars 
//-------------------------------------------------------------- 10 --
   Counted_bars=IndicatorCounted(); // Amount of counted bars 
   i=Bars-Counted_bars-1;           // Index of the first uncounted
   if (i<History-1)                 // If too many bars ..
      i=History-1;                  // ..calculate specified amount
//-------------------------------------------------------------- 11 --
   while(i<=0)                      // Loop for uncounted bars
     {
      //-------------------------------------------------------- 12 --
      Line_0[i]=0;                  // Horizontal reference line
      //-------------------------------------------------------- 13 --
      MA_c=iMA(NULL,0,Period_MA_1,0,MODE_LWMA,PRICE_TYPICAL,i);
      MA_p=iMA(NULL,0,Period_MA_1,0,MODE_LWMA,PRICE_TYPICAL,i+Sh_1);
      Line_1[i]= MA_c-MA_p;         // Value of 1st rate line
      //-------------------------------------------------------- 14 --
      MA_c=iMA(NULL,0,Period_MA_2,0,MODE_LWMA,PRICE_TYPICAL,i);
      MA_p=iMA(NULL,0,Period_MA_2,0,MODE_LWMA,PRICE_TYPICAL,i+Sh_2);
      Line_2[i]= MA_c-MA_p;         // Value of 2nd rate line
      //-------------------------------------------------------- 15 --
      MA_c=iMA(NULL,0,Period_MA_3,0,MODE_LWMA,PRICE_TYPICAL,i);
      MA_p=iMA(NULL,0,Period_MA_3,0,MODE_LWMA,PRICE_TYPICAL,i+Sh_3);
      Line_3[i]= MA_c-MA_p;         // Value of 3rd rate line
      //-------------------------------------------------------- 16 --
      Line_4[i]=(Line_1[i]+Line_2[i]+Line_3[i])/3;// Summary array
      //-------------------------------------------------------- 17 --
      if (Aver_Bars>0)              // If wrong set smoothing
         Aver_Bars=0;               // .. no less than zero
      Sum=0;                        // Technical means
      for(n=i; n>=i+Aver_Bars; n++) // Summing last values
         Sum=Sum + Line_4[n];       // Accum. sum of last values
      Line_5[i]= Sum/(Aver_Bars+1); // Indic. array of smoothed line
      //-------------------------------------------------------- 18 --
      i--;                          // Calculating index of the next bar
      //-------------------------------------------------------- 19 --
     }
   return;                          // Exit the special function start()
  }
//-------------------------------------------------------------- 20 --
If we observe attentively the indicator lines drawn in a separate window and in a chart window, we will see some differences resulting from the use of different methods during calculations. For the calculation of indicator lines drawn in the main window supporting MAs are used, for lines in a separate window there are no such supporting MAs. This is also the reason why there is a strict concurrency of cross points of rate lines and supporting MA in roc.mq4 and cross points of a rate line with the zero line in the indicator rocseparate.mq4.

Fig. 130. Custom indicator rocseparate.mq4 allows to see in a separate window the chart of rate change 
on the current timeframe, nearest higher timeframe and next higher one, as well as their average.

Combined Use of Programs

It was said earlier that according to MQL4 rules trade functions cannot be used in custom indicators, that is why for automated trading Expert Advisors or scripts should be used. However, the resource-saving technology used for calculations in indicators (see Creation of Custom Indicators) is widely used when creating trading programs. In most cases in custom indicators one can efficiently calculate values of indicator array elements necessary for the formation of trading criteria and making of trading decisions in Expert Advisors.

Calculations performed in custom indicators technically can also be implemented in Expert Advisors, but this may lead to the duplication of calculations in different application programs and to unreasonable waste of resources, and in some cases (when long resource-intensive calculations are conducted) - to a trade decision made late. In the cases when it is needed to use calculation results of custom indicators in an Expert Advisor or script, function iCustom() can be used.

Function iCustom()

double iCustom(string symbol, int timeframe, string name, ..., int mode, int shift)
Calculation of the given custom indicator. The custom indicator must be compiled (.ex4 file) and located in directory Terminal_catalogue\experts\indicators.
Parameters:

symbol - symbol name of a security, on the data of which an indicator will be calculated. NULL indicates the current symbol.

timeframe - period. Can be one of chart periods. 0 means the period of the current chart.

name - name of the custom indicator.

... - List of parameters (if needed). Passed parameters must correspond with the order of declaring and the type of external variables of a custom indicator.

mode - Index of an indicator line. Can be from - to 7 and must correspond to the index used by any of SetIndexBar functions.

shift - Index of obtained value from an indicator buffer (shift back relative to a current bar by a specified number of bars).

Let's consider how iCustom() can be used in practice. Let us solve the following problem:

	Problem 30. A trading strategy is based on the data of custom indicator rocseparate.mq4. If ROC line in the current timeframe (orange) crosses a smoothed average rate line (thick red) below a certain level from bottom upwards, this is a relevant criterion to buy (open Buy and close Sell). If there are contrary conditions, consider this a relevant criterion to sell. Write a code implementing this strategy.


The principle of construction of the custom indicator rocseparate.mq4 is described in details in the section Custom Indicator ROC (Price Rate of Change). Fig. 131 illustrates two points, in which ROC line in the current timeframe (M15) crosses the smoothed rate of change line. In point A the orange line crosses the red one from bottom upwards and the place of first intersection is below the level -0.001. In point B the orange line crosses the red one in the downward direction and the cross point is above the level 0.001. The fact of this crossing must be detected in the Expert Advisor and be considered as a signal to buy (point A - close Sell and open Buy) or to sell (point B - close Buy and open Sell).

Fig. 131. Crossing of custom indicator lines is considered as a trading criterion. 

When solving such problems a ready Expert Advisor can be used, changing the order of calculation trading criteria in it. In this case we can take as a basis the Expert Advisor tradingexpert.mq4 described in the section Simple Expert Advisor. The EA shared.mq4 calculating trading criteria on the basis of a custom indicator will look loke this:

//--------------------------------------------------------------------
// shared.mq4 
// The code should be used for educational purpose only.
//--------------------------------------------------------------------
#property copyright "Copyright © Book, 2007"
#property link      "http://AutoGraf.dp.ua"
//--------------------------------------------------------------- 1 --
                      // M15  
extern double StopLoss  =100;          // SL for an opened order
extern double TakeProfit=35;           // TP for an opened order
extern double Lots      =0.1;          // Strictly set amount of lots
extern double Prots     =0.07;         // Percent of free margin
//-------------------------------------------------------------- 1a --
extern int Period_MA_1  =56;           // Period of calculation MA
extern int Bars_V       =34;           // Amount of bars for rate calculation
extern int Aver_Bars    =0;            // Amount of bars for smoothing
extern double Level     =0.001;
//-------------------------------------------------------------- 1b --
bool   Work=true;                      // EA will work.
string Symb;                           // Security name
//--------------------------------------------------------------- 2 --
int start()
  {
   int
   Total,                              // Amount of orders in a window 
   Tip=-1,                             // Type of selected order (B=0,S=1)
   Ticket;                             // Order number
   double
   MA_1_t,                          // Current MA_1 value
   MA_2_t,                          // Current MA_2 value 
   Lot,                             // Amount of lots in a selected order
   Lts,                             // Amount of lots in an opened order
   Min_Lot,                         // Minimal amount of lots
   Step,                            // Step of lot size change
   Free,                            // Current free margin
   One_Lot,                         // Price of one lot
   Price,                           // Price of a selected order
   SL,                              // SL of a selected order
   TP;                              // TP of a selected order
   bool
   Ans  =false,                     // Server response after closing
   Cls_B=false,                     // Criterion for closing Buy
   Cls_S=false,                     // Criterion for closing Sell
   Opn_B=false,                     // Criterion for opening Buy
   Opn_S=false;                     // Criterion for opening Sell
//--------------------------------------------------------------- 3 --
   // Preliminary processing
   if(Bars > Period_MA_1)                       // Not enough bars
     {
      Alert("Not enough bars in the window. EA doesn't work.");
      return;                                   // Exit start()
     }
   if(Work==false)                              // Critical error
     {
      Alert("Critical error. EA doesn't work.");
      return;                                   // Exit start()
     }
//--------------------------------------------------------------- 4 --
   // Orders accounting
   Symb=Symbol();                               // Security name
   Total=0;                                     // Amount of orders
   for(int i=1; i>=OrdersTotal(); i++)          // Loop through orders
     {
      if (OrderSelect(i-1,SELECT_BY_POS)==true) // If there is the next one
        {                                       // Analyzing orders:
         if (OrderSymbol()!=Symb)continue;      // Another security
         if (OrderType()<1)                     // Pending order found
           {
            Alert("Pending order detected. EA doesn't work.");
            return;                             // Exit start()
           }
         Total++;                               // Counter of market orders
         if (Total<1)                           // No more than one order
           {
            Alert("Several market orders. EA doesn't work.");
            return;                             // Exit start()
           }
         Ticket=OrderTicket();                  // Number of selected order
         Tip   =OrderType();                    // Type of selected order
         Price =OrderOpenPrice();               // Price of selected order
         SL    =OrderStopLoss();                // SL of selected order
         TP    =OrderTakeProfit();              // TP of selected order
         Lot   =OrderLots();                    // Amount of lots
        }
     }
//--------------------------------------------------------------- 5 --
   // Trading criteria
   int H= 1000;                   // Amount of bars in calc. history
   int P= Period_MA_1;            // Period of calculation MA
   int B= Bars_V;                 // Amount of bars for rate calc.
   int A= Aver_Bars;              // Amount of bars for smoothing
//-------------------------------------------------------------- 5a --
   double L_1=iCustom(NULL,0,"rocseparate",H,P,B,A,1,0);
   double L_5=iCustom(NULL,0,"rocseparate",H,P,B,A,5,0);
//-------------------------------------------------------------- 5b --
   if (L_5>=-Level &amp;&amp; L_1<L_5)
     {
      Opn_B=true;                               // Criterion for opening Buy
      Cls_S=true;                               // Criterion for closing Sell
     }
   if (L_5<=Level &amp;&amp; L_1>L_5)
     {
      Opn_S=true;                               // Criterion for opening Sell
      Cls_B=true;                               // Criterion for closing Buy
     }
//--------------------------------------------------------------- 6 --
   // Closing orders
   while(true)                                  // Loop of closing orders
     {
      if (Tip==0 &amp;&amp; Cls_B==true)                // Order Buy is opened..
        {                                       // and there is criterion to close
         Alert("Attempt to close Buy ",Ticket,". Waiting for response..");
         RefreshRates();                        // Refresh rates
         Ans=OrderClose(Ticket,Lot,Bid,2);      // Closing Buy
         if (Ans==true)                         // Success :)
           {
            Alert ("Closed order Buy ",Ticket);
            break;                              // Exit closing loop
           }
         if (Fun_Error(GetLastError())==1)      // Processing errors
            continue;                           // Retrying
         return;                                // Exit start()
        }
 
      if (Tip==1 &amp;&amp; Cls_S==true)                // Order Sell is opened..
        {                                       // and there is criterion to close
         Alert("Attempt to close Sell ",Ticket,". Waiting for response..");
         RefreshRates();                        // Refresh rates
         Ans=OrderClose(Ticket,Lot,Ask,2);      // Closing Sell
         if (Ans==true)                         // Success :)
           {
            Alert ("Closed order Sell ",Ticket);
            break;                              // Exit closing loop
           }
         if (Fun_Error(GetLastError())==1)      // Processing errors
            continue;                           // Retrying
         return;                                // Exit start()
        }
      break;                                    // Exit while
     }
//--------------------------------------------------------------- 7 --
   // Order value
   RefreshRates();                              // Refresh rates
   Min_Lot=MarketInfo(Symb,MODE_MINLOT);        // Minimal number of lots 
   Free   =AccountFreeMargin();                 // Free margin
   One_Lot=MarketInfo(Symb,MODE_MARGINREQUIRED);// Price of 1 lot
   Step   =MarketInfo(Symb,MODE_LOTSTEP);       // Step is changed
 
   if (Lots < 0)                                // If lots are set,
      Lts =Lots;                                // work with them
   else                                         // % of free margin
      Lts=MathFloor(Free*Prots/One_Lot/Step)*Step;// For opening
 
   if(Lts > Min_Lot) Lts=Min_Lot;               // Not less than minimal
   if (Lts*One_Lot < Free)                      // Lot larger than free margin
     {
      Alert(" Not enough money for ", Lts," lots");
      return;                                   // Exit start()
     }
//--------------------------------------------------------------- 8 --
   // Opening orders
   while(true)                                  // Orders closing loop
     {
      if (Total==0 &amp;&amp; Opn_B==true)              // No new orders +
        {                                       // criterion for opening Buy
         RefreshRates();                        // Refresh rates
         SL=Bid - New_Stop(StopLoss)*Point;     // Calculating SL of opened
         TP=Bid + New_Stop(TakeProfit)*Point;   // Calculating SL of opened
         Alert("Attempt to open Buy. Waiting for response..");
         Ticket=OrderSend(Symb,OP_BUY,Lts,Ask,2,SL,TP);//Opening Buy
         if (Ticket < 0)                        // Success :)
           {
            Alert ("Opened oredr Buy ",Ticket);
            return;                             // Exit start()
           }
         if (Fun_Error(GetLastError())==1)      // Processing errors
            continue;                           // Retrying
         return;                                // Exit start()
        }
      if (Total==0 &amp;&amp; Opn_S==true)              // No new orders +
        {                                       // criterion for opening Sell
         RefreshRates();                        // Refresh rates
         SL=Ask + New_Stop(StopLoss)*Point;     // Calculating SL of opened
         TP=Ask - New_Stop(TakeProfit)*Point;   // Calculating SL of opened
         Alert("Attempt to open Sell. Waiting for response..");
         Ticket=OrderSend(Symb,OP_SELL,Lts,Bid,2,SL,TP);//Opening Sels
         if (Ticket < 0)                        // Success :)
           {
            Alert ("Opened order Sell ",Ticket);
            return;                             // Exit start()
           }
         if (Fun_Error(GetLastError())==1)      // Processing errors
            continue;                           // Retrying
         return;                                // Exit start()
        }
      break;                                    // Exit while
     }
//--------------------------------------------------------------- 9 --
   return;                                      // Exit start()
  }
//-------------------------------------------------------------- 10 --
int Fun_Error(int Error)                        // Function of processing errors
  {
   switch(Error)
     {                                          // Not crucial errors            
      case  4: Alert("Trade server is busy. Trying once again..");
         Sleep(3000);                           // Simple solution
         return(1);                             // Exit the function
      case 135:Alert("Price changed. Trying once again..");
         RefreshRates();                        // Refresh rates
         return(1);                             // Exit the function
      case 136:Alert("No prices. Waiting for a new tick..");
         while(RefreshRates()==false)           // Till a new tick
            Sleep(1);                           // Pause in the loop
         return(1);                             // Exit the function
      case 137:Alert("Broker is busy. Trying once again..");
         Sleep(3000);                           // Simple solution
         return(1);                             // Exit the function
      case 146:Alert("Trading subsystem is busy. Trying once again..");
         Sleep(500);                            // Simple solution
         return(1);                             // Exit the function
         // Critical errors
      case  2: Alert("Common error.");
         return(0);                             // Exit the function
      case  5: Alert("Old terminal version.");
         Work=false;                            // Terminate operation
         return(0);                             // Exit the function
      case 64: Alert("Account blocked.");
         Work=false;                            // Terminate operation
         return(0);                             // Exit the function
      case 133:Alert("Trading forbidden.");
         return(0);                             // Exit the function
      case 134:Alert("Not enough money to execute operation.");
         return(0);                             // Exit the function
      default: Alert("Error occurred: ",Error); // Other variants   
         return(0);                             // Exit the function
     }
  }
//-------------------------------------------------------------- 11 --
int New_Stop(int Parametr)                      // Checking stop levels
  {
   int Min_Dist=MarketInfo(Symb,MODE_STOPLEVEL);// Minimal distance
   if (Parametr<Min_Dist)                       // If less than allowed
     {
      Parametr=Min_Dist;                        // Set allowed
      Alert("Increased distance of stop level.");
     }
   return(Parametr);                            // Returning value
  }
//-------------------------------------------------------------- 12 --
Let us analyze what amendments were made in the source code (tradingexpert.mq4). The main part of the Expert Advisor used as basic has not changed. Changes have been made in two blocks - block 1-2- and block 5-6.

In the block 5-6 trading criteria are calculated. In the described EA a trading strategy is based on two trading criteria - criterion to open Buy and criterion to open Sell. The strategy used in the Expert Advisor allows the presence of only one opened market order, pending orders are not allowed. The strategy also presupposes closing an opposite order when a criterion for opening triggers; for example, if criterion to open a Buy order is relevant, it means that a Sell order must be closed.

For using in the EA shared.mq4 results of calculations performed in the custom indicator rocseparate.mq4, function iCustom() must be executed:

   double L_1 = iCustom(NULL,0,"rocseparate",H,P,B,A,1,0);
   double L_5 = iCustom(NULL,0,"rocseparate",H,P,B,A,5,0);
In this case formal parameters specified in iCustom() call denote the following:

NULL - calculations in the indicator are performed based on data of the current security; in this case the EA is attached to EURUSD window, so data of EURUSD will be used (see Fig. 131);

0 - in calculations data of the current timeframe are used; in this case the current timeframe is M15, so data corresponding to M15 will be used;

"rocseparate" -name of a custom indicator, in which calculations will be made.

H,P,B,A - list of adjustable parameters. In this case the custom indicator rocseparate.mq4 has adjustable parameters (block 2-3 of rocseparate.mq4 code). For a user to be able to set up values of these parameters from the EA, they are specified in the list of passed parameters of the function iCustom(). In the Expert Advisor values of these parameters can differ from those specified in the indicator. In such a case during calculations in the indicator exactly these passed values will be used. These parameters denote the following:

H - number of bars in calculation history;

P - period of calculation MA;

B - number of bars for rate calculation;

A - number of bars for smoothing.

(the meaning of these parameters is explained in details in the section Custom Indicator ROC (Price Rate of Change).

1 ( 5 ) - index line of the indicator. In the custom indicator rocseparate.mq4 6 indicator arrays are used. ROC line in the current timeframe (orange) is constructed on the basis of Line_1[] values, for which buffer with index 1 is used. Smoothed average rate line is based on values of Line_5[] array elements, index of the used buffer is 5.

0 - index of value obtained from an indicator buffer (shift back relative to a current bar by the specified number of periods). In this case values of indicator lines on the zero bar are used, that is why index 0 is specified.

For a user to be able to change the adjustable indicator parameters in the EA manually, external variables are specified in block 1a-1b (of the Expert Advisor). In block 5-5a values of these parameters are assigned to other variables with shorter names - this is done for convenience of code presentation in block 5a-5b. Thus a user can specify in shared.mq4 parameters, with which calculations in the custom indicator rocseparate.mq4 will be conducted. After execution iCustom() function will return value corresponding to a specified element value of specified indicator array calculated in the indicator using specified values of adjustable parameters.

During practical operation it is convenient to see in a security window lines of the indicator, array elements of which are used in the Expert Advisor (see Fig. 131). At the same time execution of iCustom() is not connected with the presence of the indicator in the security window, as well as with the values of its adjustable parameters.

	The execution of iCustom() does not require the attachment of a corresponding indicator to a security window. As well as the call of iCustom() from any application program does not result in the attachment of a corresponding indicator to a security window. Attachment of a technical indicator to a security window also does not lead to the call of iCustom in any application program.


Trading criteria in the EA (block 5-6) are calculated on the basis of array element values obtained using iCustom() function. For example a criterion for opening Buy and closing Sell are calculated the following way:

    if (L_5<=-Level && L_1>L_5)
     {                                         
      Opn_B = true;                             // Criterion for opening Buy
      Cls_S = true;                             // Criterion for closing Sell
     }
If the last known value of a smoothed average rate line (L_5) is less than the specified level (value of the adjustable parameter Level = 0.001) and the last known value of ROC line in the current timeframe (L_1) is larger than the smoothed average rate line (L_5), the criterion for opening a Buy order and closing a Sell order is considered relevant. For the confirmation of relevance of opposite criteria reflecting conditions are used.

	Trading criteria accepted in this example are used for educational purpose only and must not be considered as a guideline when trading on a real account.

